



DECUS

PROGRAM LIBRARY

DECUS NO.	8-522
TITLE	'PAGEIT'
AUTHOR	William R. Anderson, Jr.
COMPANY	Portsmouth Abbey School Portsmouth, Rhode Island
DATE	March 15, 1972
SOURCE LANGUAGE	PAL III

DECUS

WORLDWIDE PRINTING



/ VERSION 2.0
/ MARCH 15, 1972 + + + ASSEMBLED ON PDP 8/E.
/ PAL-III OVERLAY FOR FORMATTED PAGING IN THE
/ FIRST AND THIRD PASS WITH PAGE NUMBERS.
/ BY : WILLIAM ANDERSON, JR.,
/ PORTSMOUTH ABBEY SCHOOL, PORTSMOUTH RHODE ISLAND

/ THIS ROUTINE REQUIRES 200 LOCATIONS (OCTAL) AT *7200.

/ BY MEANS OF PATCHING PAL-III, IT IS POSSIBLE
/ TO MONITOR ITS OUTPUT , SO THAT A COUNT
/ ON LINE FEEDS CAN BE MAINTAINED.
/ THIS ROUTINE PROVIDES THE OPTION OF SELECTING
/ A FORMATTED OUTPUT ON THE TERMINAL.
/ ALL PRINTED MATTER WILL PLACED ON FORM SIZE PAGES
/ WITH PAGE NUMBERS AT THE TOP.
/ THE SIZE OF EACH PAGE CAN BE SET BY THE USER OR
/ WILL BE DEFAULTED TO 55 LINES.
/ AN EJECT OPTION IS GIVEN TO THE USER SO THAT
/ HE CAN EJECT TO A NEW PAGE WHEN DESIRED.

/ OPERATION : LOAD PAL-III, USING BIN LOADER.
/ LOAD THE PAGING OVERLAY.
/

/ YOU CAN USE PAL-III NORMALLY, EXCEPT YOU CALL
/ THE PAGING ROUTINE BY SETTING BIT '8' TO ONE ON THE
/ CONSOLE. IF BIT '8' IS NOT SET, THEN PAL-III WILL
/ FUNCTION NORMALLY WITHOUT PAGING.
/ REMEMBER ON THE SECOND PASS, YOU MUST LOWER BIT '8'
/ SO THAT THE HEADER WILL NOT APPEAR ON YOUR OBJECT TAPE.
/ TO USE THE EJECT OPTIONS, PLACE AN UP ARROW IN YOUR
/ SOURCE LISTING AFTER A SLASH, OR IF YOU ARE USING A
/ DEC WRITER , TYPE THE OVERHEAD BRACKET.
/ UPON RECIEVING THE EJECT COMMAND, THE PAGE ROUTINE
/ WILL LINE FEED TO A NEW PAGE.
/ WHEN PAL-III HALTS, THE PAGE ROUTINE WILL COMPLETE
/ THE UNFINISHED PAGE WITH LINE FEEDS.
/ THE TARE OFF POINT OF EACH PAGE IS PRINTED BY A '-'.
/ IF THE USER TARES IT OFF AT THAT POINT, WHILE THE OUTPUT
/ IS BEING PRINTED THEN HE CAN BE ASSURED OF UNIFIED PAGE
/ LENGTHS.

/ THE USER CAN SET HIS OWN PAGE LENGTHS BY CHANGING
/ LOCATION 7362 TO THE NEGATIVE OF THE NUMBER OF LINES
/ HE WANTS PER PAGE. THE USER MUST SET THE NUMBER IN
/ OCTAL AND COMPLEMENT AND INCREMENT IT.

/ RESTRICTIONS : THIS ROUTINE REQUIRES ONE PAGE OF
/ PAL'S SYMBOL TABLE STORAGE, SO THAT THE
/ MAXIMUM OF 576 SYMBOLS IS REDUCED TO 544.
/ EJECT COMMAND --> ;

/ PATCHES TO PAL-III.

```
*0
0000 7200 INIT, RESET / ADDRESS OF POINTER RESET ROUTINE.

*214
0214 5400 RETURN, JMP I INIT / ON PAL-III CALL FOR OSR, RESET
/ POINTERS FOR PAGES AND LINES.

*240
0240 5764 CONT2, JMP I FORM / ON HALT, COMPLETE PAGE.

*362
0362 5763 JMP I .+1 / REPLACE TTY PRINTING ROUTINE WITH
0363 7204 CHECK / PAGING MONITOR.
0364 7225 FORM, FEED / ADDRESS FOR HALT PROCESSING.
/ EJECT COMMAND --> †
```

/ MAIN ROUTINE = ONE PAGE IN LENGTH

```

*7200
7200 3354 RESET, DCA PAGE / ZERO PAGE NUMBER
7201 4233 JMS NEWP / IF ROUTINE CALLED, PROCESS HEADER.
7202 7604 LAS
7203 5770 JMP I K215 / RETURN ADDRESS TO PAL-III.
7204 4324 CHECK, JMS OUT / PRINT A CHARACTER.
7205 1361 TAD M212 / CHECK FOR LINE FEED.
7206 7450 SNA
7207 5217 JMP LINES
7210 1356 TAD M124 / CHECK FOR UP ARROW, EJECT OPTION.
7211 7640 SZA CLA
7212 5215 JMP GOHOME / IF NOT, RETURN.
7213 4331 SKIP, JMS EJECT / PROCESS LIFT ON PAGE.
7214 4233 JMS NEWP
7215 5616 GOHOME, JMP I.+1 / RETURN TO PAL-III.
7216 0365 FORM+1
7217 1362 LINES, TAD LINE
7220 1363 TAD M65 / CHECK FOR ONE PRINTED PAGE.
7221 7700 K7700, SMA CLA
7222 5213 JMP SKIP
7223 2362 ISZ LINE / INCREMENT LINE COUNT.
7224 5215 JMP GOHOME
7225 4331 FEED, JMS EJECT / COMPLETE PAGE ON HALT.
7226 1364 TAD DASH
7227 4324 JMS OUT / PRINT DASH FOR TARE POINT.
7230 7602 GETOUT, HLT CLA / NORMAL PAL-III HALT.
7231 5632 JMP I .+1 / RETURN TO PAL-III.
7232 0241 CONT2+1
/ EJECT COMMAND --> †

```

/ SUBROUTINES

7233	0000	NEWP,	0	/ PAGE HEADER ROUTINE.
7234	7604		LAS	
7235	0357		AND K10	/ CHECK BIT 8. FOR PAGE OPTION.
7236	7650		SNA CLA	
7237	5312		JMP HOME	/ IF NOT, RETURN.
7240	2354		ISZ PAGE	/ INCREMENT PAGE COUNT.
7241	3362		DCA LINE	/ ZERO LINE COUNT.
7242	1364		TAD DASH	/ PRINT TARE OFF POINT.
7243	4324		JMS OUT	
7244	7300		CLA CLL	
7245	4313		JMS RETUP	
7246	1365		TAD M5	
7247	3355		DCA REGSET	/ TEMPORARY STORAGE.
7250	1372		TAD BUFF	
7251	3331		DCA EJECT	
7252	1731		TAD I EJECT	/ PRINT PAGE LABEL.
7253	4324		JMS OUT	
7254	7200		CLA	
7255	2331		ISZ EJECT	
7256	2355		ISZ REGSET	
7257	5252		JMP --5	
7260	1354		TAD PAGE	
7261	0357		AND K10	/ PEEL OUT SECOND DIGIT
7262	7650		SNA CLA	
7263	5274		JMP NUMPRI	/ PRINT NUMBER ON NON OVERFLOW.
7264	1354		TAD PAGE	
7265	0360		AND K2	
7266	7650		SNA CLA	
7267	5274		JMP NUMPRI	
7270	1354		TAD PAGE	
7271	0221		AND K7700	/ INCREMENT FIRST DIGIT AND CLEAR / THE SECOND DIGIT.
7272	1366		TAD K100	
7273	3354		DCA PAGE	
7274	1354	NUMPRI,	TAD PAGE	
7275	0221		AND K7700	/ PEEL OFF FIRST DIGIT.
7276	7012	K12,	RTR	
7277	7012		RTR	
7300	7012		RTR	/ SWAP AROUND INTO BITS 6-11. / ADD BASE
7301	1373		TAD BASE	
7302	4324		JMS OUT	
7303	7200		CLA	
7304	1354		TAD PAGE	/ PEEL OUT SECOND / AND PRINT IT.
7305	0367		AND K17	
7306	1373		TAD BASE	
7307	4324		JMS OUT	
7310	7200		CLA	
7311	4313		JMS RETUP	
7312	5633	HOME,	JMP I NEWP	

7313	0000	RETUP,	0	/ ROUTINE TO LF 4 TIMES AND RETURN.
7314	1370		TAD K215	
7315	4324		JMS OUT	
7316	7200		CLA	
7317	1371		TAD M4	
7320	3355		DCA REGSET	
7321	1276		TAD K12	
7322	4346		JMS MULTIP	
7323	5713		JMP I RETUP	
7324	0000	OUT,	0	/ PRINTING ROUTINE.
7325	6046		TLS	
7326	6041		TSF	
7327	5326		JMP .-1	
7330	5724		JMP I OUT	
7331	0000	EJECT,	0	
7332	7604		LAS	
7333	0357		AND K10	/ CHECK BIT 8. FOR PAGE OPTION.
7334	7650		SNA CLA	
7335	5345		JMP GOODBYE	
7336	4313		JMS RETUP	/ CLEAR AND RESET PAGE.
7337	1363		TAD M65	
7340	1365		TAD M5	
7341	1362		TAD LINE	/ COMPUTE NUMBER OF LINES LEFT.
7342	3355		DCA REGSET	/ TEMPORARY STORAGE.
7343	1276		TAD K12	
7344	4346		JMS MULTIP	
7345	5731	GOODBYE,	JMP I EJECT	
7346	0000	MULTIP,	0	
7347	4324		JMS OUT	/ PRINT CHAR.
7350	2355		ISZ REGSET	
7351	5347		JMP .-2	
7352	7200		CLA	
7353	5746		JMP I MULTIP	
		/ EJECT COMMAND --> :		

/ DECLARES

7354	0000	PAGE,	0
7355	0000	REGSET,	0
7356	7654	M124,	-124
7357	0010	K10,	10
7360	0002	K2,	2
7361	7566	M212,	-212
7362	0000	LINE,	0
7363	7713	M65,	-65
7364	0255	DASH,	255
7365	7773	M5,	-5
7366	0100	K100,	100
7367	0017	K17,	17
7370	0215	K215,	215
7371	7774	M4,	-4
7372	7374	BUFF,	TEXT
7373	0260	BASE,	260
7374	0320	TEXT,	320
7375	0301		301
7376	0307		307
7377	0305		305
7400	0240	END,	240

/ END OF PAGE ROUTINE :

BASE	7373
BUFF	7372
CHECK	7204
CONT2	0240
DASH	7364
EJECT	7331
END	7400
FEED	7225
FORM	0364
GETOUT	7230
GOHOME	7215
GOODEY	7345
HOME	7312
INIT	0000
K10	7357
K100	7366
K12	7276
K17	7367
K2	7360
K215	7370
K7700	7221
LINE	7362
LINES	7217
MULTIP	7346
M124	7356
M212	7361
M4	7371
M5	7365
M65	7363
NEWP	7233
NUMPRI	7274
OUT	7324
PAGE	7354
REGSET	7355
RESET	7200
RETUP	7313
RETURN	0214
SKIP	7213
TEXT	7374

